

Extending REC with References

CS510

A New Language

- ▶ We are going to extend REC with references
- ▶ As a consequence we obtain a functional language that supports variables as in imperative programming
- ▶ Note that all previous programming features shall still be available, including
 - ▶ let expressions
 - ▶ procedures
 - ▶ recursion

Implicit vs Explicit References

There are essentially two ways of doing this:

1. Treat every variable as a mutable reference
 - ▶ The resulting language is called IMPLICIT-REFS
 - ▶ References are implicit
2. Add mutable references to the non-mutable ones
 - ▶ The resulting language is called EXPLICIT-REFS
 - ▶ References are explicit

We are going to study both

Implicit vs Explicit References

IMPLICIT-REFS

```
1 let g = let count = 0
2         in proc(d) {
3             begin set count = count + 1;
4                   count
5             end }
6 in (g 11) - (g 22)
```

EXPLICIT-REFS

```
1 let g = let counter = newref(0)
2         in proc (d) {
3             begin
4                 setref(counter, deref(counter) + 1);
5                 deref(counter)
6             end }
7 in (g 11) - (g 22)
```

Implicit References

Concrete and Abstract Syntax

Expressed Values

The Store

The Interpreter

Explicit References

Concrete and Abstract Syntax

Concrete Syntax

Two new productions that are added to those of REC

$$\begin{aligned}\langle Expression \rangle &::= \text{set } \langle Identifier \rangle = \langle Expression \rangle \\ \langle Expression \rangle &::= \text{begin } \langle Expression \rangle^{+(:)} \text{ end}\end{aligned}$$

- ▶ The `set` expression is assignment
- ▶ A `begin ... end` expression evaluates its subexpressions in order and returns the value of the last one.

Concrete Syntax – An Example

```
1 let g = let count = 0
2         in proc(d) {
3             begin
4                 set count = count + 1;
5                 count
6             end
7         }
8 in (g 11) - (g 22)
```

IMPLICIT-REFS: Abstract Syntax

```
1  type expr =  
2    | Var of string  
3    | Int of int  
4    | Add of expr*expr  
5    | Sub of expr*expr  
6    | Mul of expr*expr  
7    | Div of expr*expr  
8    | Let of string*expr*expr  
9    | IsZero of expr  
10   | ITE of expr*expr*expr  
11   | Proc of string*expr  
12   | App of expr*expr  
13   | Letrec of string*string*expr*expr  
14   | Set of string*expr  
15   | BeginEnd of expr list
```


Example - Abstract Syntax

```
1 let g = let count = 0
2       in proc(d) {
3           begin
4               set count = count + 1;
5               count
6           end
7       }
8 in (g 11) - (g 22)
```

```
1 AProg
2   (Let ("g",
3       Let ("count", Int 0,
4           Proc ("d", BeginEnd [Set ("count", Add (Var "count"
5               ↪ ", Int 1)); Var "count"])),
6       Sub (App (Var "g", Int 11), App (Var "g", Int 22))))
```

Implicit References

Concrete and Abstract Syntax

Expressed Values

The Store

The Interpreter

Explicit References

Concrete and Abstract Syntax

Unit Value

- ▶ An expression in REC can return one of:

$\text{ExpVal} = \text{Int} + \text{Bool} + \text{Proc}$

- ▶ What should the result of this be?

```
let x = 2 in set x = 7
```

- ▶ Same question: what is the result of evaluating a `set` expression?

`set` is evaluated to cause an effect (modify the store/memory), not return a result

Unit Value

- ▶ We introduce a special value for this situation, namely `UnitVal`

$$\text{ExpVal} = \text{Int} + \text{Bool} + \text{Proc} + \text{Unit}$$

- ▶ `set` evaluates to a unit value

Implicit References

Concrete and Abstract Syntax

Expressed Values

The Store

The Interpreter

Explicit References

Concrete and Abstract Syntax

Motivating the Store

Consider this assignment statement in Java

$$x := x + 1$$

- ▶ As we know, variables have two readings:
 - ▶ An address (the blue occurrence of x)
 - ▶ A value (the red occurrence of x)
- ▶ Environments will now associate **only references** to variables
- ▶ A *reference* is an expression of the form

$$\text{RefVal } n$$

for n an integer

Motivating the Store

Consider this assignment statement in Java

$$x := x + 1$$

- ▶ Environments will now associate **only references** to variables
- ▶ Revisiting the two readings of a variable
 - ▶ Blue x : we just look it up in the environment
 - ▶ Red x : we look it up in the environment and then use that location to get its value in a **store** or **memory**
- ▶ Our interpreter thus shall need a **store**

Environment and Store

Environment

RefVal 0	RefVal 1	RefVal 2
x	y	z

Store

NumVal 3	BoolVal true	NumVal 7
0	1	2

Summing Up Our Analysis

$$\begin{aligned}\text{ExpVal} &= \text{Int} + \text{Bool} + \text{Proc} + \text{Unit} \\ \text{DenVal} &= \text{Ref}(\text{ExpVal})\end{aligned}$$

- ▶ Recall: Denoted values = Values assigned to variables in the environment
- ▶ References exist only as the bindings of variables.
- ▶ Two options for the type of the interpreter
 1. Store is passed as argument

`eval_expr :: env -> store -> expr -> exp_val`

2. Store is held in global variable (*we choose this one*)

`eval_expr :: env -> expr -> exp_val`

Implementing the Store

- ▶ There are many ways to represent the store (eg. Lists, User-defined datatypes, etc.)
 - ▶ We choose arrays since it is simple
 - ▶ The references will just be indices
 - ▶ An example

NumVal 3	BoolVal true	NumVal 7	NumVal 28
0	1	2	3

- ▶ Sample operations we should support
 - ▶ Create a store
 - ▶ Lookup the value associated to a reference
 - ▶ Ask the store for a fresh reference and assign it a value (allocation)

Store Interface - store.mli

```
1  type 'a t (* t is the (opaque) type of the store *)
2
3  val empty_store : int -> 'a -> 'a t
4
5  val new_ref : 'a t -> 'a -> int
6
7  val deref : 'a t -> int -> 'a
8
9  val set_ref : 'a t -> int -> 'a -> unit
10
11 val store_to_list : 'a t -> 'a list
```

Implementing the Store

```
1 type 'a t = {mutable data: 'a array;  
2             mutable size: int}  
3 (* data is declared mutable so the store may be  
4    ↪ resized *)  
5 let empty_store i v = { data=Array.make i v; size=0 }  
6  
7 let get_size st = st.size
```

Implementing the Store

```
1  let enlarge_store st value =  
2    let new_array = Array.make (st.size*2) value  
3    in Array.blit st.data 0 new_array 0 st.size;  
4    st.data<-new_array  
5  
6  let new_ref st value =  
7    if Array.length (st.data)=st.size  
8    then enlarge_store st value  
9    else ();  
10   begin  
11     st.data.(st.size)<-value;  
12     st.size<-st.size+1;  
13     st.size-1  
14   end
```

Implementing the Store

NumVal 3	BoolVal true	NumVal 7
0	1	2

After newref (NumVal 28):

NumVal 3	BoolVal true	NumVal 7	NumVal 28
0	1	2	3

Implementing the Store

```
1  let deref st ref = st.data.(ref)
2
3  let set_ref st ref value =
4      if ref >= st.size
5      then failwith "Index out of bounds"
6      else st.data.(ref) <- Value
7
8  let rec take n = function
9      | [] -> []
10     | x::xs when n > 0 -> x::take (n-1) xs
11     | x::xs -> []
12
13  let store_to_list st =
14      take st.size @@ Array.to_list @@ st.data
```

Implementing the Store

NumVal 3	BoolVal true	NumVal 7	NumVal 28
0	1	2	3

After `setref st 1 (NumVal 42)`:

NumVal 3	NumVal 42	NumVal 7	NumVal 28
0	1	2	3

Implicit References

Concrete and Abstract Syntax

Expressed Values

The Store

The Interpreter

Explicit References

Concrete and Abstract Syntax

Specification of Behavior of the Interpreter

- ▶ We now specify how the interpreter `eval_expr` behaves

- ▶ Input:

- ▶ expression
- ▶ environment
- ▶ store

- ▶ Output:

- ▶ expressed value
- ▶ `updated` store

```
eval_expr :: env -> store -> expr -> (expval * store)
```

- ▶ Note: As mentioned, in our implementation the store won't be passed as a parameter, it will be held in a global variable

Specification of Behavior of Interpreter on Constants

- ▶ First we revisit the interpreter's behavior for the simplest cases.

```
eval_expr  $\rho$   $\sigma$  (Int n) = (NumVal n,  $\sigma$ )
```

- ▶ ρ is an environment
- ▶ σ is the store
- ▶ Int *var* is the expression to evaluate
- ▶ In this case the store σ is returned unaltered

Specification of Behavior of Interpreter on Difference

```
1 eval_expr  $\rho$   $\sigma_0$  (Sub exp1 exp2) =  
2   let (val1,  $\sigma_1$ ) = eval_expr  $\rho$   $\sigma_0$  exp1  
3   in let (val2,  $\sigma_2$ ) = eval_expr  $\rho$   $\sigma_1$  exp2  
4   in NumVal ((expval_to_num val1) -  
5             (expval_to_num val2)),  $\sigma_2$ )
```

Specification of Behavior of Interpreter on Variables

1. look up the identifier in the environment to find the location to which it is bound
2. look up in the store to find the value at that location

```
eval_expr  $\rho$   $\sigma$  (Var var) = ( $\sigma(\rho(var))$ ,  $\sigma$ )
```

Specification of Behavior of Interpreter w.r.t. Assignment

```
1 eval_expr  $\rho$   $\sigma_0$  (Set (var, exp1)) =  
2   let (val1,  $\sigma_1$ ) = eval_expr  $\rho$   $\sigma_0$  exp1  
3   in (UnitVal, [ $\rho$ (var) = val1] $\sigma_1$ )
```

- ▶ UnitVal is the value of an expression, such as set, whose sole objective is to cause an effect
- ▶ [ρ (var) = val1] σ_1 stands for “update the reference ρ (var) in the store σ_1 with the new value val1

Implementation

```
1 eval_expr  $\rho$   $\sigma$  (Var var) = ( $\sigma(\rho(var))$ ), $\sigma$ )
```

```
1 eval_expr (en:env) (e:expr) :exp_val =  
2   match e with  
3   | Int n          -> NumVal n  
4   | Var id         ->  
5     (match apply_env en id with  
6      | None -> failwith @@ "Variable "^id^" undefined"  
7      | Some ev -> Store.deref g_store @@ refVal_to_int  
          ↪ ev)
```

Implementation

```
1 eval_expr  $\rho$   $\sigma_0$  (Set (var,exp1)) =  
2   let (val1, $\sigma_1$ ) = eval_expr  $\rho$   $\sigma_0$  exp1  
3   in (NumVal 27, [ $\rho$ (var) = val1] $\sigma_1$ )
```

```
1 eval_expr (en:env) (e:expr) :exp_val =  
2   match e with  
3   ...  
4   | Set(id,e) ->  
5     let v=eval_expr en e  
6     in Store.set_ref g_store (refVal_to_int (from_some  
7       ↪ (apply_env g_store en id))) v;  
8     UnitVal
```

- We'll explain later why `apply_env` needs the store

Updating the Implementation of Extant REC Features

- ▶ The implementation of features that were already present in REC also have to be updated
- ▶ We've already seen the cases for constants and difference
- ▶ Here is the specification of the behavior of the interpreter for `let`

```
1 eval_expr  $\rho$   $\sigma_0$  (Let(var,exp1,body))=  
2   let (val1, $\sigma_1$ ) = eval_expr  $\rho$   $\sigma_0$  exp1  
3   in eval_expr [var = l] $\rho$  [l = val1] $\sigma_1$  body
```

- ▶ `l` denotes a fresh store location

Updating the Implementation of Extant REC Features

► The implementation for `let`

```
1 eval_expr (en:env) (e:expr) :exp_val =  
2   match e with  
3     ...  
4   | Let(x, e1, e2) ->  
5     let v1 = eval_expr en e1  
6     in let l = Store.new_ref g_store v1  
7     in eval_expr (extend_env en x (RefVal l)) e2
```

Specification of Behavior of Interpreter w.r.t. Procedures

```
1 eval_expr  $\rho$   $\sigma$  (Proc(var, body)) =  
2   (procVal(var, body,  $\rho$ ),  $\sigma$ )
```

- No changes w.r.t. REC here

Specification of Behavior of Interpreter w.r.t. Procedure Calls

```
1 eval_expr  $\rho$   $\sigma_0$  (App(rator,rand)) =  
2   let (proc, $\sigma_1$ ) = eval_expr  $\rho$   $\sigma_0$  rator  
3     (arg, $\sigma_2$ ) = eval_expr  $\rho$   $\sigma_1$  rand  
4   in (apply_proc proc arg,  $\sigma_2$ )
```

- No changes w.r.t. REC here but must update apply_proc

Specifying the Behavior of Procedure Application

```
1  apply_proc (ProcVal(var, body,  $\rho$ )) val  $\sigma$  =  
2      eval_expr [var = !] $\rho$  [! = val] $\sigma$  body
```

- Here *!* denotes a fresh store location

Implementing Procedure Application

```
1 (* exp_val -> exp_val -> exp_val *)
2 let rec apply_proc f a =
3   match f with
4     ProcVal (x,b,env) ->
5       let l = Store.new_ref g_store a
6       in eval_expr (extend_env env x (RefVal l)) b
7   | _ -> failwith "apply_proc: Not a procVal"
```

Environment Lookup

- ▶ Last we have to update environment lookup
- ▶ This operation was called `apply_env`
- ▶ Only the case for `letrec` has to be updated
- ▶ Before doing so we recall its definition

apply-env as Implemented in REC

```
1  let rec apply_env (env:env) (id:string):exp_val option
    ↪   =
2    match env with
3    | EmptyEnv -> None
4    | ExtendEnv (key,value,env) ->
5      if id=key
6      then Some value
7      else apply_env env id
8    | ExtendEnvRec (key,param,body,en) ->
9      if id=key
10     then Some (ProcVal (param,body,env))
11     else apply_env en id
```

- Note: `apply_env` would return a closure

Updating `apply-env`

- ▶ Before:
 - ▶ `apply-env` would return a closure
- ▶ Now:
 - ▶ `apply-env` should return a **reference** to a location in the **store** containing the appropriate closure
- ▶ The reason: how the interpreter manages variable lookup

```
1 eval_expr  $\rho$   $\sigma$  (Var var) = ( $\sigma(\rho(\textit{var}))$ ),  $\sigma$ )
```

$\rho(\textit{var})$ is environment lookup of *var* in ρ

Updating `apply-env`

```
1  let rec apply_env (st:exp_val Store.t) (env:env) (id:
    ↪ string):exp_val option =
2  match env with
3  | EmptyEnv -> None
4  | ExtendEnv (key,value,env) ->
5      if id=key
6      then Some value
7      else apply_env st env id
8  | ExtendEnvRec (key,param,body,en) ->
9      if id=key
10     then Some (RefVal (Store.new_ref st (ProcVal(param
    ↪ ,body,env))))
11     else apply_env st en id
```

- ▶ The closure is stored in the store: we return a reference to it
- ▶ This completes the implementation of IMPLICIT-REFS.

The Interpreter for IMPLICIT-REFS

- ▶ Code available in Canvas Modules/Interpreters
- ▶ Directory `implicit-refs`
- ▶ Compile with `ocamlbuild -use-menhir interp.ml`
- ▶ Make sure the `.ocamlinit` file is in the folder of your sources
- ▶ Run `utop`

Implicit References

Concrete and Abstract Syntax

Expressed Values

The Store

The Interpreter

Explicit References

Concrete and Abstract Syntax

EXPLICIT-REFS: A language with explicit references

1. We now define a new language EXPLICIT-REFS, which adds references as expressed values to our language.
2. Concrete and Abstract Syntax
3. Specification
4. Implementation

Implicit vs Explicit References

IMPLICIT-REFS

```
1 let g = let count = 0
2         in proc(d)
3             begin { set count = count + 1;
4                     count
5                     end }
6 in (g 11) - (g 22)
```

EXPLICIT-REFS

```
1 let g = let counter = newref(0)
2         in proc (d) {
3             begin
4                 setref(counter, deref(counter) +1);
5                 deref(counter)
6             end }
7 in (g 11) - (g 22)
```

Implicit vs Explicit Store

- ▶ In the implicit store design, **every variable is mutable**.
 - ▶ Allocation, dereferencing and mutation are built into the language.
- ▶ The explicit reference design gives a clear account of allocation, dereferencing, and mutation
 - ▶ All these operations are explicit in the programmer's code.

Expressed and Denoted Values

Before (IMPLICIT-REFS)

$$\begin{aligned}\text{ExpVal} &= \text{Int} + \text{Bool} + \text{Proc} + \text{Unit} \\ \text{DenVal} &= \text{Ref}(\text{ExpVal})\end{aligned}$$

Now (EXPLICIT-REFS):

$$\begin{aligned}\text{ExpVal} &= \text{Int} + \text{Bool} + \text{Proc} + \text{Unit} + \text{Ref}(\text{ExpVal}) \\ \text{DenVal} &= \text{ExpVal}\end{aligned}$$

Think of the result of evaluating:

```
1 newref(7)
```

Or

```
1 let x = newref(7) in x
```


Expressed and Denoted Values

Before (IMPLICIT-REFS)

$$\begin{aligned}\text{ExpVal} &= \text{Int} + \text{Bool} + \text{Proc} + \text{Unit} \\ \text{DenVal} &= \text{Ref}(\text{ExpVal})\end{aligned}$$

Now (EXPLICIT-REFS):

$$\begin{aligned}\text{ExpVal} &= \text{Int} + \text{Bool} + \text{Proc} + \text{Unit} + \text{Ref}(\text{ExpVal}) \\ \text{DenVal} &= \text{ExpVal}\end{aligned}$$

- ▶ Before: programs could only produce numbers, booleans, closures or unit as a result
- ▶ Now: programs can **also** produce references as a result (or even store them)

Expressed Values for EXPLICIT-REFS

► Expressed values before

```
1 type exp_val =  
2   | NumVal of int  
3   | BoolVal of bool  
4   | ProcVal of string*Ast.expr*env
```

► Now

```
1 type exp_val =  
2   | NumVal of int  
3   | BoolVal of bool  
4   | ProcVal of string*Ast.expr*env  
5   | UnitVal  
6   | RefVal of int
```

Environment and Store

Environment

NumVal 3	NumVal 3	RefVal 0
x	y	z

Store

NumVal 7	RefVal 9	BoolVal true
0	1	2

Yet Another Example – References to References

```
1 let x = newref(newref(0))
2 in begin
3     setref(deref(x), 11);
4     deref(deref(x))
5 end
```

- ▶ Allocates a new reference containing 0.
- ▶ Then binds `x` to a reference containing the above reference.
- ▶ The value of `deref(x)` is a reference to the first reference.
- ▶ So when it evaluates the `setref`, it is the first reference that is modified, and the entire program returns 11.

EXPLICIT-REFS: Concrete Syntax

$\langle Expression \rangle ::= \langle Number \rangle$
 $\langle Expression \rangle ::= \langle Expression \rangle - \langle Expression \rangle$
 $\langle Expression \rangle ::= \text{zero? } (\langle Expression \rangle)$
 $\langle Expression \rangle ::= \text{if } \langle Expression \rangle$
 then $\langle Expression \rangle$ else $\langle Expression \rangle$
 $\langle Expression \rangle ::= \langle Identifier \rangle$
 $\langle Expression \rangle ::= \text{let } \langle Identifier \rangle = \langle Expression \rangle \text{ in } \langle Expression \rangle$
 $\langle Expression \rangle ::= \text{proc } (\langle Identifier \rangle) \{ \langle Expression \rangle \}$
 $\langle Expression \rangle ::= (\langle Expression \rangle \langle Expression \rangle)$
 $\langle Expression \rangle ::= \text{letrec } \langle Identifier \rangle (\langle Identifier \rangle) = \langle Expression \rangle$
 in $\langle Expression \rangle$

EXPLICIT-REFS: Concrete Syntax

$\langle Expression \rangle ::= \text{newref } (\langle Expression \rangle)$
 $\langle Expression \rangle ::= \text{deref } (\langle Expression \rangle)$
 $\langle Expression \rangle ::= \text{setref } (\langle Expression \rangle)$
 $\langle Expression \rangle ::= \text{begin } \langle Expression \rangle^{+(:)} \text{ end}$

We add four new operations

- ▶ **newref**: allocates a new location and returns a reference to it.
- ▶ **deref**: dereferences a reference: that is, it returns the contents of the location that the reference represents.
- ▶ **setref**: changes the contents of the location that the reference represents.
- ▶ **begin...end**, as before.

EXPLICIT-REFS: Abstract Syntax

```
1  type expr =  
2    | Var of string  
3    | Int of int  
4    | Add of expr*expr  
5    | Sub of expr*expr  
6    | Mul of expr*expr  
7    | Div of expr*expr  
8    | Let of string*expr*expr  
9    | IsZero of expr  
10   | ITE of expr*expr*expr  
11   | Proc of string*expr  
12   | App of expr*expr  
13   | Letrec of string*string*expr*expr  
14   | BeginEnd of expr list  
15   | NewRef of expr  
16   | DeRef of expr  
17   | SetRef of expr*expr
```

Example

```
1 let x = newref(newref(0))
2 in begin
3     setref(deref(x), 11);
4     deref(deref(x))
5 end
```

```
1 AProg
2   (Let ("x", NewRef (NewRef (Int 0))),
3     BeginEnd [SetRef (DeRef (Var "x"), Int 11);
4               DeRef (DeRef (Var "x"))]))
```


Specification – `newref`

```
1 eval_expr  $\rho$   $\sigma_0$  (NewRef exp) =  
2   let (val,  $\sigma_1$ ) = eval_expr  $\rho$   $\sigma_0$  exp  
3   in (RefVal l, [l=val] $\sigma_1$ )
```

- ▶ l is fresh, that is $l \notin \text{dom}(\sigma_1)$
- ▶ RefVal is the tag that indicates that the result is a reference

Implementation — `newref`

```
1 eval_expr  $\rho$   $\sigma_0$  (NewRef exp) =  
2   let (val,  $\sigma_1$ ) = eval_expr  $\rho$   $\sigma_0$  exp  
3   in (RefVal l, [l=val] $\sigma_1$ )
```

```
1 eval_expr (en:env) (e:expr) :exp_val =  
2   match e with  
3   ...  
4   | NewRef(e) ->  
5       RefVal(Store.new_ref g_store (eval_expr en e))
```

Specification – `deref` and `setref`

```
1 eval_expr  $\rho$   $\sigma_0$  (DeRef exp) =  
2   let (RefVal l,  $\sigma_1$ ) = eval_expr  $\rho$   $\sigma_0$  exp  
3   in ( $\sigma_1(l)$ ,  $\sigma_1$ )
```

```
1 eval_expr  $\rho$   $\sigma_0$  (SetRef exp1 exp2) =  
2   let (RefVal l,  $\sigma_1$ ) = eval_expr  $\rho$   $\sigma_0$  exp1  
3       (val,  $\sigma_2$ ) = eval_expr  $\rho$   $\sigma_1$  exp2  
4   in (UnitVal, [l=val] $\sigma_2$ )
```

Implementation — deref

```
1 eval_expr  $\rho$   $\sigma_0$  (DeRef exp) =  
2   let (RefVal l,  $\sigma_1$ ) = eval_expr  $\rho$   $\sigma_0$  exp  
3   in ( $\sigma_1(l)$ ,  $\sigma_1$ )
```

```
1 eval_expr (en:env) (e:expr) :exp_val =  
2   match e with  
3   ...  
4   | DeRef(e) ->  
5     let v1 = eval_expr en e  
6     in Store.deref g_store (refVal_to_int v1)
```

Implementation — setref

```
1 eval_expr  $\rho$   $\sigma_0$  (SetRef exp1 exp2) =  
2   let (RefVal l, $\sigma_1$ ) = eval_expr  $\rho$   $\sigma_0$  exp1  
3     (val, $\sigma_2$ ) = eval_expr  $\rho$   $\sigma_1$  exp2  
4   in (UnitVal,[l=val] $\sigma_2$ )
```

```
1 eval_expr (en:env) (e:expr) :exp_val =  
2   match e with  
3   ...  
4   | SetRef(e1,e2) ->  
5     let v1=eval_expr en e1  
6     in let v2=eval_expr en e2  
7     in Store.set_ref g_store (refVal_to_int v1) v2;  
8     UnitVal
```

The Interpreter for EXPLICIT-REFS

- ▶ Code available in Canvas Modules/Interpreters
- ▶ Directory `explicit-refs`
- ▶ Compile with `ocamlbuild -use-menhir interp.ml`
- ▶ Make sure the `.ocamlinit` file is in the folder of your sources
- ▶ Run `utop`